

Creating a name plate with a raised image or raised text.

Before you start, you need to determine your background (shape or image) and your foreground (shape, image, or text).



Basic shapes:

Here are some basic shapes you may want to use:

- `Disk[{0,0},5]` is a filled-in circle centered at (0,0) with radius 5.
- `Disk[{0,0},{6,4}]` is a filled-in ellipse centered at (0,0) with major axis 6 and minor axis 4.
- `RegularPolygon[n]` is a filled-in regular n-gon.
- `Rectangle[{x1,y1},{x2,y2}]` is a rectangle with corners (x1,y1), (x1,y2), (x2,y2), and (x2,y1).

Then transform it into a mesh using

```
shape = DiscretizeGraphics[Graphics[<the shape you chose here>]]
```

Using an image:

If you want to use an image as your foreground or background, drag and drop it into Mathematica and give it a name, like this:

```
flower =  ;
```

(If your image has grayscale or colors, you will probably want to use `Binarize`.)

Transform it into a mesh using:

```
image= ImageMesh[ColorNegate[flower]]
```

Using text:

If you want to use text as your foreground, update your text, font, and font size:

```
word = Text[  
    Style["word", Bold, FontFamily -> "Bauhaus 93", FontSize -> 50]]
```

Discretize it using:

```
text = DiscretizeGraphics[word, _Text]
```

Putting it all together:

Choose which of these things is your foreground and background, and use `RegionResize` to make sure that the foreground fits into the background: (You may need to modify the constants.)

```
resizedShape = RegionResize[shape, 1.3]
resizedImage = RegionResize[image, 1]
```

We are going to need to move these regions to have the same center:

```
translatedShape = TransformedRegion[
    resizedShape,
    TranslationTransform[
        -Map[Mean, RegionBounds[resizedShape]]
    ]
]
translatedImage = TransformedRegion[
    resizedImage,
    TranslationTransform[
        -Map[Mean, RegionBounds[resizedImage]]
    ]
]
```

Display them both together to make sure they are overlapping the right way:

```
Show[translatedShape, translatedImage]
```

If so, you're ready to continue. Define the background and the foreground:

```
background = translatedShape
foreground = translatedImage
```

Use *Mathematica* to determine the background minus the foreground, and the boundaries:

```
middle = BoundaryDiscretizeRegion[RegionDifference[background, foreground]]
outerboundary = RegionBoundary[background];
innerboundary = RegionBoundary[foreground];
```

Now construct all the polygons for the name plate. Adjust the values of the heights as necessary.

```
nameplate = Show[{
    RegionProduct[background, Point[{{0.}}]],
    RegionProduct[outerboundary, Line[{{0.}, {0.1}}]],
    RegionProduct[middle, Point[{{0.1}}]],
    RegionProduct[innerboundary, Line[{{0.1}, {0.2}}]],
    RegionProduct[foreground, Point[{{0.2}}]]
}]
```

Last, export your file to STL:

```
Export[NotebookDirectory[] <> "nameplate.stl", nameplate]
```